

Grafikë themelorë në Maxima

Qëllimet dhe objektivat

- Komandad grafike në Maxima
- Krijimi i grafikëve dydimensionalë
- Krijimi i grafikëve tredimensionalë

Përmbajtja

1 Komanda plot2d

2 Komanda plot3d

Komanda plot2d

- Forma më e thjeshtë e komandës plot2d:

```
plot2d(exp(-t/10)*cos(10*t), [t,0,10]);
```

- Në wxMaxima një grafik mund të afishohet inline me anë të komandës wxplot2d:

```
wxplot2d(exp(-t/10)*cos(10*t), [t,0,10]);
```

- Specifikimi i rangut vertikal:

```
plot2d(1/x, [x,0.1,2]);
```

```
plot2d(1/x, [x,0.1,2], [y,0,5]);
```

```
plot2d(sec(x), [x,-2,2]);
```

```
plot2d(sec(x), [x,-2,2], [y,-20,20]);
```

Opione të komandës plot2d

- Vizatimi i më tepër se një lakoreje:

```
f(x):=exp(-x/10)*cos(10*x);  
g(x):=0.5*exp(-x/10)*sin(10*x);  
wxplot2d([f(x),g(x)], [x,0,4]);
```

- Specifikimet e legjendës dhe labelave:

```
h(x):=0.5*g(x);  
wxplot2d([f(x),g(x),h(x)], [x,0,4],  
[legend, "pozita", "shpejtësia", "nxitim");  
wxplot2d(sin(x), [x,-%pi,%pi],  
[xlabel, "koha (s)"],  
[ylabel, "shpejtësia (m/s)"]);  
wxplot2d([sin(x),sec(x),tan(x)], [x,-5,5], [y,-5,5],  
[legend, "sin", "sec", "tan"],  
[xlabel, "këndi"], [ylabel, "funksionet"]);
```

Paraqitja grafike e të dhënave diskrete

- Paraqitja grafike e të dhënave diskrete:

```
x: [0,1,2,3,4,5,6];  
y: [0, 1.8, 3.5, 10.5, 12.0, 15.5, 17.3];  
wxplot2d([discrete, x, y]);  
wxplot2d([discrete, x, y], [style, [points]]);
```

- Opcioni points për style mund të ketë deri në tre opçione shtesë që përcaktojnë rrezen ngjyrën dhe formën e pikave:

```
wxplot2d([discrete, x, y], [style, [points]]);  
wxplot2d([discrete, x, y], [style, [points, 1]]);  
wxplot2d([discrete, x, y], [style, [points, 10]]);  
wxplot2d([discrete, x, y], [style, [points, 3, 2]]);  
wxplot2d([discrete, x, y], [style, [points, 3, 6]]);  
wxplot2d([discrete, x, y], [style, [points,3,1,2]]);  
wxplot2d([discrete, x, y], [style, [points,6,4,10]]);
```

Paraqitja grafike e të dhënave diskrete (Vazhdim)

- Opciona tjetra për style:

```
wxplot2d([discrete, x, y], [style, [lines,5,3]]);  
wxplot2d([discrete, x, y], [style, [lines,10,7]]);  
wxplot2d([discrete, x, y], [style, [linespoints,5,3,3,2]]);  
wxplot2d([discrete, x, y], [style, [linespoints,1,3,3,2]]);
```

- Paraqitja grafike e disa tipe të dhëash:

```
wxplot2d([1.5*t+sin(t), [discrete, x, y]], [t,0,10],  
[style, [lines,1,2], [points, 3,1,11]]);  
wxplot2d([1.5*t+sin(t), [discrete, x, y]], [t,0,10],  
[style, [lines,2,7], [points, 2,2,9]],  
[legend, "vija", "pika"]);  
wxplot2d([[discrete, x, y], [discrete, x, y]],  
[t,0,10], [style, [lines,1,2], [points, 4,6,10]],  
[legend, "vija", "pika"]);
```

Vizatimi i grafikëve parametrikë

- Vizatimi i grafikëve parametrikë:

```
wxplot2d([parametric, cos(t), sin(t)], [t,-%pi,%pi]);
wxplot2d([parametric, cos(t), sin(t)], [t,-%pi,%pi],
          same_xy);
wxplot2d([[parametric, 5*cos(t), 16*sin(t)],
           [discrete, x, y]],
           [t,0,%pi/2], [style, [lines,2,5], [points,4,1,7]],
           [legend, "parametrik", "diskret"]);
wxplot2d([2.5*(t-sin(t)),
           [parametric, 5*cos(t/6), 16*sin(t/6)],
           [discrete, x, y]],
           [t,0,10],
           [style, [lines,1,2], [lines,2,5], [points,4,1,7]],
           [legend, "funkcion", "parametrik", "diskret"]);
```

Komanda plot3d

- Komanda plot3d mund të përdoret për të paraqitur grafikisht sipërfaqe që janë grafikë funksionesh të formës $z = f(x, y)$:

```
kill(x,y);  
plot3d(x*sin(y)+y*sin(x), [x,-2*pi,2*pi],  
[y,-2*pi,2*pi]);
```

- Në wxMaxima një grafik tredimensional mund të afishohet inline me anë të komandës wxplot3d:

```
wxplot_size;  
wxplot3d(x*sin(y)+y*sin(x), [x,-2*pi,2*pi],  
[y,-2*pi,2*pi]);  
wxplot_size: [400,250];  
wxplot3d(x*sin(y)+y*sin(x), [x,-2*pi,2*pi],  
[y,-2*pi,2*pi]);
```

Opcione të komandës plot3d

- Opcioni `grid` përcakton numrin e pikave për ndryshoret x dhe y në paraqitjen grafike. Vlera e nënkuptuar është `[grid,30,30]`:

```
wxplot3d(x*sin(y)+y*sin(x), [x,-2*pi,2*pi],  
          [y,-2*pi,2*pi], [grid,10,10]);  
wxplot3d(x*sin(y)+y*sin(x), [x,-2*pi,2*pi],  
          [y,-2*pi,2*pi], [grid,50,50]);
```

- Paraqitja grafike e sipërfaqeve parametrike:

```
plot3d([cos(u)*(2+v*cos(u/2)), sin(u)*(2+v*cos(u/2)),  
        v*sin(u/2)],  
       [u,-pi,pi], [v,0,1], [grid,50,15]);
```

Opcione të komandës plot3d (Vazhdim)

- Opcioni gnuplot_preamble:

```
wxplot3d(x*sin(y)+y*sin(x), [x,-2*pi,2*pi],  
[y,-2*pi,2*pi],  
[gnuplot_preamble, "set view map"]);  
plot2d(th+sin(th), [th,0,2*pi],  
[gnuplot_preamble, "set polar"]);  
plot3d(1, [th,0,2*pi], [ph,-pi/2,pi/2],  
[gnuplot_preamble, "set mapping spherical"]);  
plot3d(1, [th,0,2*pi], [ph,-pi/2,pi/2],  
[gnuplot_preamble, "set mapping spherical"], same_xyz);  
plot3d(sin(th)*cos(ph), [th,0,2*pi], [ph,-pi/2,pi/2],  
[grid,20,20],  
[gnuplot_preamble, "set mapping spherical"]);  
plot3d(1, [th,0,2*pi], [z,-1,1],  
[gnuplot_preamble, "set mapping cylindrical"], same_xyz);  
plot3d(z, [th,0,2*pi], [z,0,5],  
[gnuplot_preamble, "set mapping cylindrical"]);
```

Transformimi i koordinatave

- Opcioni transform_xy:

```
wxplot3d([r,th,r*th], [r,0,2], [th,0,%pi], [grid,10,10]);  
wxplot3d([r,th,r*th], [r,0,2], [th,0,%pi], [grid,10,10],  
[transform_xy,  
 make_transform([r,th,z],r*cos(th),r*sin(th),z]]);  
wxplot3d(sqrt(1-x^2-y^2), [x,-1,1], [y,-1,1]);  
wxplot3d([1,th,ph], [th,0,2*%pi], [ph,0,%pi/2],  
[transform_xy,  
 make_transform([r,th,ph],  
 r*cos(th)*cos(ph),r*sin(th)*cos(ph),r*sin(ph))]);
```

Paraqitja grafike e lakoreve tredimensionale

- Për të paraqitur grafikisht një lakore tredimensionale mund të shfrytëzohet paraqitja parametrike me ndryshorën e dytë dummy:

```
wxplot3d([cos(t),sin(t),t], [t,0,6*pi], [s,0,6*pi],  
[grid,50,50]);  
plot3d([th,th/12,1], [th,-6*pi,6*pi],  
[ph,-%pi/2,%pi/2],  
[grid,120,1], same_xyz,  
[gnuplot_preamble, "set mapping spherical"]);
```

Përfundim

- Zbatimi i komandave grafike në Maxima për krijimin e grafikëve dydimensionale
- Lakoret dydimensionale diskrete dhe parametrike
- Zbatimi i komandave grafike në Maxima për krijimin e grafikëve tredimensionalë
- Sipërfaqet dhe lakoret tredimensionale parametrike
- Paraqitja grafike në koordinata polare, cilindrike dhe sferike